Sage and Linear Algebra Worksheet

FCLA Section SS

Robert Beezer
Department of Mathematics and Computer Science
University of Puget Sound

Fall 2019

1 Vector Spaces

It is easy in Sage to make a reasonable facsimile of $\mathbf{C}^n$. We just restrict our attention to rational entries rather than complex entries. This vector space contains vectors with 4 slots, each filled with a rational number.

```
V = QQ^4
V
```

Demonstration 1 We can test membership using the word/command `in`. Try vectors with different numers of slots, and perhaps include the complex number `2 + 3*I` as an entry.

2 Vector Form of Solutions to Homogeneous Systems

These are the coefficient matrix and vector of constants from yesterday's big system that led to a colored matrix in reduced row-echelon form.

The `.right_kernel()` method will give the vectors of the vector form of the solutions to a homogeneous system when used with the `basis='pivot'` option.

```
A = matrix(QQ, [[  1,  2, 12,   1,  13,  5,  2],
                [ -2, -3, -21,  0, -13,  2, -5],
                [  1,  3,  15,  4,  28, 25,  0],
                [ -2, -3, -21, -1, -15, -6, -3],
                [  1,  1,   9,  1,   4,  9,  1]])
b = vector(QQ, [8, -15, 7, -10, 3])
```

```
A.right_kernel(basis='pivot')
```

Rows of the "basis matrix" are vectors in yesterday's linear combination (with scalars x_3, x_5, x_6). This is a spanning set for the null space of the matrix A. See Theorem VFSLS and Theorem SSNS.

Theorem PSPHS can explain how to use a single solution to the non-homogeneous system and the spanning set of the null space of the coefficient matrix to arrive at all solutions to the system. Here is a single solution to the system.

```
A.solve_right(b)
```

Notice that this vector is the solution when we set each free variable to zero, which is the "other" vector from yesterday that is not part of the linear combination.

3 Spanning Sets

Example ABS from FCLA.

```
x1 = vector(QQ,[1,1,3,1])
x2 = vector(QQ,[2,1,2,-1])
x3 = vector(QQ,[7,3,5,-5])
x4 = vector(QQ,[1,1,-1,2])
x5 = vector(QQ,[-1,0,9,0])
W = span([x1, x2, x3, x4, x5])
W
```

Demonstration 2 Make a "random" linear combination of the five vectors and test for membership (which will be trivially true, repeatedly). Remember to use the `*` operator for vector scalar multiplication.

But not any old vector is in W.

```
v = vector(QQ, [1, 1, -3, 2])
v in W
```

It should make sense that arbitrary linear combinations are in the span. How did we manufacture a vector *not* in the span? Stay tuned.